# Maintainer:  Bartłomiej Piotrowski <bpiotrowski@archlinux.org>
# Contributor: Andrea Scarpino <andrea@archlinux.org>
# Contributor: Tobias Powalowski <tpowa@archlinux.org>

pkgname=fsarchiver
pkgver=0.8.7
pkgrel=1
pkgdesc='Safe and flexible file-system backup and deployment tool'
arch=('x86_64')
url='https://www.fsarchiver.org/'
license=('GPL2')
depends=('bzip2' 'e2fsprogs' 'lzo' 'xz' 'libgcrypt' 'zlib' 'lz4' 'zstd')
source=(https://github.com/fdupoux/fsarchiver/releases/download/$pkgver/fsarchiver-$pkgver.tar.gz)
sha256sums=('f7cd059e70fc57e888db282c622ec050c0dcdbaacc65e3c1eb163cd9d92d810d')

build() {
  cd $pkgname-$pkgver
  ./configure --prefix=/usr --sbindir=/usr/bin
  make
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="$pkgdir" install
}
